
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 8.10 which is part of the ‘IMS_34229_IWD_13wk43’ test suite in ATS ‘iwd-EUTRA-B2012-03_D13wk43’.

2. Table of Contents
101 Jan – 31 Dec 2013


21. Overview


22. Table of Contents


23. Corrections required


2Change 1 – f_IMS_SubscribeRequest_MessageHeaderRX


5Change 2 – f_IMS_Register_Notify


6Change 3 – a_SystemInterfaceHandler




3. Corrections required
This section describes the TTCN changes required to ‘iwd-EUTRA-B2012-03_D13wk43’ release.
Change 1 – f_IMS_SubscribeRequest_MessageHeaderRX
	Function name
	f_IMS_SubscribeRequest_MessageHeaderRX

	Reason for change
	According to RFC 3261, port 5060 is the default port only, and the UE can choose a different port as well.

	Summary of change
	Replaced hard-coded number by any number.

	TTCN module
	Common/IMS/IMS_Procedures_Registration

	MCC160 Comment
	Accepted


Before change

  function f_IMS_SubscribeRequest_MessageHeaderRX(EventType p_EventType := "reg",

                                                  template SipUrl p_SipUrlTo := omit,

                                                  template (present) charstring p_ExpireTime := "600000",

                                                  boolean p_SessionIdSupported := false) runs on IMS_PTC return template (present) MessageHeader

  { /* p_EventType:

     *   "reg"             - normal case acc. A.1.4

     *   "conference"      - "SUBSCRIBE for conference event package" acc. annex A.5.1; NOTE in this case p_SipUrlTo shall be present and set acc. to "px_FinalConferenceUri"

     *   "message-summary" - "SUBSCRIBE for Message Waiting Indication package" acc. annex A.6.1 */

    /* @sic R5-130750 change 4: specific IMPU for messages sent in DL sic@ */

    /* @sic R5s130XXX: p_IsGIBA removed sic@ */

    var template (present) MessageHeader v_MessageHeader_Subscribe := cr_MessageHeader_Dummy;

    var IMS_ProtectedPorts_Type v_Protected;

    var template (present) PAccessNetworkInfo v_PAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *);

    var template (present) SipUrl v_SipUrlFrom := f_IMS_PTC_ImsInfo_GetSipUriSubscriptionRX();

    var template (present) SipUrl v_SipUrlTo;

    if (ispresent(p_SipUrlTo)) {

      v_SipUrlTo := p_SipUrlTo;         // NOTE: compiler warning cannot be avoided here

    } else {

      v_SipUrlTo := v_SipUrlFrom;

    }

    v_MessageHeader_Subscribe.cSeq        := cr_CseqDef(?, "SUBSCRIBE");

    v_MessageHeader_Subscribe.fromField   := cr_FromWithTag(v_SipUrlFrom);

    v_MessageHeader_Subscribe.toField     := cr_ToDef(v_SipUrlTo);

    v_MessageHeader_Subscribe.expires     := cr_Expires(p_ExpireTime);  // @sic R5-130548 sic@

    v_MessageHeader_Subscribe.callId      := cr_CallId(?);

    v_MessageHeader_Subscribe.sessionId   := f_IMS_SessionId_RX_IfSupported(p_SessionIdSupported);   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Subscribe.maxForwards := cr_MaxForwardsDef;

    v_MessageHeader_Subscribe.event       := cr_EventDef(p_EventType);

    if (f_IMS_PTC_SecurityScheme_IsGiba()) {

      v_MessageHeader_Subscribe.route   := cr_RouteSubs (f_IMS_PTC_Pcscf_Get(), *, px_IMS_Scscf);

      v_MessageHeader_Subscribe.via     := cr_ViaDef(-, cr_HostPort(?));                                       // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.contact := cr_Contact(cr_SipUri_HostPort(?, tsc_IMS_PortNumber_5060));         // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.require := omit;

      v_MessageHeader_Subscribe.proxyRequire := omit;

      v_MessageHeader_Subscribe.securityVerify := omit;

      v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo ifpresent;

    }

    else {

      v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

      v_MessageHeader_Subscribe.route   := cr_RouteSubs (f_IMS_PTC_Pcscf_Get(), v_Protected.Port_ps, px_IMS_Scscf);

      v_MessageHeader_Subscribe.via     := cr_ViaDef(-, cr_HostPort(?, v_Protected.Port_us));                   // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.contact := cr_Contact(cr_SipUri_HostPort(?, v_Protected.Port_us));              // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.require := cr_RequireSecAgree;

      v_MessageHeader_Subscribe.proxyRequire := cr_ProxyRequireDef;

      v_MessageHeader_Subscribe.securityVerify := f_IMS_PTC_BuildSecurityVerifyHeader();

      v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo

    }

    select (p_EventType) {

      case ("reg") {                // normal case

        v_MessageHeader_Subscribe.accept := cr_AcceptDef(c_xmlreginfoAplication) ifpresent;

      }

      case ("conference") {         // "SUBSCRIBE for conference event package" acc. annex A.5.1

        v_MessageHeader_Subscribe.accept := cr_AcceptDef("application/conference-info+xml");

        v_MessageHeader_Subscribe.expires := cr_Expires(?);

      }

      case ("message-summary") {    // "SUBSCRIBE for Message Waiting Indication package" acc. annex A.6.1

        v_MessageHeader_Subscribe.accept := cr_AcceptDef("application/simple-message-summary")

      }

    }

    return v_MessageHeader_Subscribe;

  }
After change

  function f_IMS_SubscribeRequest_MessageHeaderRX(EventType p_EventType := "reg",

                                                  template SipUrl p_SipUrlTo := omit,

                                                  template (present) charstring p_ExpireTime := "600000",

                                                  boolean p_SessionIdSupported := false) runs on IMS_PTC return template (present) MessageHeader

  { /* p_EventType:

     *   "reg"             - normal case acc. A.1.4

     *   "conference"      - "SUBSCRIBE for conference event package" acc. annex A.5.1; NOTE in this case p_SipUrlTo shall be present and set acc. to "px_FinalConferenceUri"

     *   "message-summary" - "SUBSCRIBE for Message Waiting Indication package" acc. annex A.6.1 */

    /* @sic R5-130750 change 4: specific IMPU for messages sent in DL sic@ */

    /* @sic R5s130XXX: p_IsGIBA removed sic@ */

    var template (present) MessageHeader v_MessageHeader_Subscribe := cr_MessageHeader_Dummy;

    var IMS_ProtectedPorts_Type v_Protected;

    var template (present) PAccessNetworkInfo v_PAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *);

    var template (present) SipUrl v_SipUrlFrom := f_IMS_PTC_ImsInfo_GetSipUriSubscriptionRX();

    var template (present) SipUrl v_SipUrlTo;

    if (ispresent(p_SipUrlTo)) {

      v_SipUrlTo := p_SipUrlTo;         // NOTE: compiler warning cannot be avoided here

    } else {

      v_SipUrlTo := v_SipUrlFrom;

    }

    v_MessageHeader_Subscribe.cSeq        := cr_CseqDef(?, "SUBSCRIBE");

    v_MessageHeader_Subscribe.fromField   := cr_FromWithTag(v_SipUrlFrom);

    v_MessageHeader_Subscribe.toField     := cr_ToDef(v_SipUrlTo);

    v_MessageHeader_Subscribe.expires     := cr_Expires(p_ExpireTime);  // @sic R5-130548 sic@

    v_MessageHeader_Subscribe.callId      := cr_CallId(?);

    v_MessageHeader_Subscribe.sessionId   := f_IMS_SessionId_RX_IfSupported(p_SessionIdSupported);   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Subscribe.maxForwards := cr_MaxForwardsDef;

    v_MessageHeader_Subscribe.event       := cr_EventDef(p_EventType);

    if (f_IMS_PTC_SecurityScheme_IsGiba()) {

      v_MessageHeader_Subscribe.route   := cr_RouteSubs (f_IMS_PTC_Pcscf_Get(), *, px_IMS_Scscf);

      v_MessageHeader_Subscribe.via     := cr_ViaDef(-, cr_HostPort(?));                                       // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.contact := cr_Contact(cr_SipUri_HostPort(?, ?));         // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.require := omit;

      v_MessageHeader_Subscribe.proxyRequire := omit;

      v_MessageHeader_Subscribe.securityVerify := omit;

      v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo ifpresent;

    }

    else {

      v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

      v_MessageHeader_Subscribe.route   := cr_RouteSubs (f_IMS_PTC_Pcscf_Get(), v_Protected.Port_ps, px_IMS_Scscf);

      v_MessageHeader_Subscribe.via     := cr_ViaDef(-, cr_HostPort(?, v_Protected.Port_us));                   // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.contact := cr_Contact(cr_SipUri_HostPort(?, v_Protected.Port_us));              // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.require := cr_RequireSecAgree;

      v_MessageHeader_Subscribe.proxyRequire := cr_ProxyRequireDef;

      v_MessageHeader_Subscribe.securityVerify := f_IMS_PTC_BuildSecurityVerifyHeader();

      v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo

    }

    select (p_EventType) {

      case ("reg") {                // normal case

        v_MessageHeader_Subscribe.accept := cr_AcceptDef(c_xmlreginfoAplication) ifpresent;

      }

      case ("conference") {         // "SUBSCRIBE for conference event package" acc. annex A.5.1

        v_MessageHeader_Subscribe.accept := cr_AcceptDef("application/conference-info+xml");

        v_MessageHeader_Subscribe.expires := cr_Expires(?);

      }

      case ("message-summary") {    // "SUBSCRIBE for Message Waiting Indication package" acc. annex A.6.1

        v_MessageHeader_Subscribe.accept := cr_AcceptDef("application/simple-message-summary")

      }

    }

    return v_MessageHeader_Subscribe;

  }
Change 2 – f_IMS_Register_Notify
	Function name
	f_IMS_Register_Notify

	Reason for change
	In case of GIBA, the SS needs to send the NOTIFY request via the unprotected port. It also waits there for the request.

	Summary of change
	Defined helper variables for the request and the response to be sent resp to arrive at unprotected port.

	TTCN module
	Common/IMS/IMS_Procedures_Registration

	MCC160 Comment
	Accepted in principle but changes are done in common modules for 34.229 test model implementation in order to keep consistency between both test models, See MCC160 implementation below.


Before change

function f_IMS_Register_Notify(SipUrl p_ContactUrl,

                                 SUBSCRIBE_Request p_SubscribeRequest,

                                 template (value) Reginfo_Type p_XmlMessage,

                                 integer p_CseqValue := 1,

                                 template (value) SubscriptionState p_SubscriptionState := cs_SubscriptionState_Active) runs on IMS_PTC

  { /* @sic R5s130756 additional changes: p_CondA2 removed sic@ */

    var template (value) MessageHeader v_MessageHeader_Notify;

    var template (present) MessageHeader v_MessageHeader_Response;

    var InternetProtocol_Type v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();

    v_MessageHeader_Notify := f_IMS_NotifyRequest_MessageHeader_A16(p_SubscribeRequest, p_CseqValue, p_SubscriptionState);

    IMS_Client.send(cas_IMS_Notify_Request(cs_IMS_RoutingInfo(v_Protocol), cs_NOTIFY_Request(p_ContactUrl, v_MessageHeader_Notify, cs_MessageBody_RegInfo(p_XmlMessage))));

    v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Notify);

    IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine200, v_MessageHeader_Response)));

  }
After change

function f_IMS_Register_Notify(SipUrl p_ContactUrl,

                                 SUBSCRIBE_Request p_SubscribeRequest,

                                 template (value) Reginfo_Type p_XmlMessage,

                                 integer p_CseqValue := 1,

                                 template (value) SubscriptionState p_SubscriptionState := cs_SubscriptionState_Active) runs on IMS_PTC

  { /* @sic R5s130756 additional changes: p_CondA2 removed sic@ */

    var template (value) MessageHeader v_MessageHeader_Notify;

    var template (present) MessageHeader v_MessageHeader_Response;

    var InternetProtocol_Type v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();


var IMS_RoutingInfo_Type v_DLRouting := cs_IMS_RoutingInfo(v_Protocol);

    var template (present) IMS_RoutingInfo_Type v_ULRouting := cr_IMS_RoutingInfo(v_Protocol);

    if (f_IMS_PTC_SecurityScheme_IsGiba())

    {

        v_DLRouting.Security := unprotected;

        v_ULRouting.Security := unprotected; 

    }
    v_MessageHeader_Notify := f_IMS_NotifyRequest_MessageHeader_A16(p_SubscribeRequest, p_CseqValue, p_SubscriptionState);

    IMS_Client.send(cas_IMS_Notify_Request(v_DLRouting, cs_NOTIFY_Request(p_ContactUrl, v_MessageHeader_Notify, cs_MessageBody_RegInfo(p_XmlMessage))));

    v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Notify);

    IMS_Client.receive(car_IMS_DATA_RSP(v_ULRouting, cr_Response(c_statusLine200, v_MessageHeader_Response)));

  }
Change 2 –MCC160Implementation 

Reason for the different implementation: The implicit assumption is, that the IMS-PTC is not aware of the transport mechanism TCP/UDP, Ipsec nor of SigComp, it just handles the SIP signalling. The management of the transport layer is done in the IP-Adaptor-PTC.  Further on, we need to take care that the implementations for the 34.229 test model and 36.523 test model harmonize.
Change 2.1
	Global type name
	IP_ADAPTOR_GlobalInfo_Type

	Reason for change
	New flag was needed  

	Summary of change
	Flag aded 

	TTCN module
	IMS/34_229_Model/IP_ADAPTOR_Component.ttcn

	MCC160 Comment
	MCC160 implementation


Before change

  type record IP_ADAPTOR_GlobalInfo_Type {

    SSPortInfo SSPortInfo optional, // Current port info(IP address, port number and protocol) at the System Simulator side to be used for IMS signalling

    boolean DLSigCompActive,

  };
After change

  type record IP_ADAPTOR_GlobalInfo_Type {

    SSPortInfo SSPortInfo optional, // Current port info(IP address, port number and protocol) at the System Simulator side to be used for IMS signalling

    boolean DLSigCompActive,

    boolean SecurityStarted  //@sic R5s130195 - Change 2
  };

Change 2.2

	Function name
	fl_SendRequestDL

	Reason for change
	Sending of requests needs to consider whether security is initated or not.

	Summary of change
	See below

	TTCN module
	IMS/34_229_Model/IP_ADAPTOR_Component.ttcn

	MCC160 Comment
	MCC160 implementation


Before change

    function fl_SendRequestDL(integer p_PdnIndex,

                            IMS_DATA_REQ p_IMS_DATA_REQ) runs on IP_ADAPTOR_PTC

  {

    var IMS_RoutingInfo_Type v_RoutingInfo := p_IMS_DATA_REQ.RoutingInfo;

    var IMS_Request_Type v_Request := p_IMS_DATA_REQ.Request;

    var template (value) ASP_DataRequest v_ASP_DataRequest;

    var template (omit) SigCompInfo v_SigCompInfo := omit;

    var MessageHeader v_MessageHeader;

    if (vc_Global.DLSigCompActive) {
      v_MessageHeader := f_IMS_Request_GetMessageHeader(v_Request);

      v_SigCompInfo := cs_SigCompDL(v_MessageHeader.callId.callid);

    }

    v_ASP_DataRequest := cas_ASP_DataRequest(v_SigCompInfo, fl_IP_ADAPTOR_GetPortInfo(v_RoutingInfo.Protocol), v_Request);

    // Requests are sent on Client ports for UDP and TCP

    if (v_RoutingInfo.Security == unprotected) {

      UnsecClient.send(v_ASP_DataRequest);

    } else {

      SecClient.send(v_ASP_DataRequest);

    }

  }

After change

  function fl_SendRequestDL(integer p_PdnIndex,

                            IMS_DATA_REQ p_IMS_DATA_REQ) runs on IP_ADAPTOR_PTC

  {

    var IMS_RoutingInfo_Type v_RoutingInfo := p_IMS_DATA_REQ.RoutingInfo;

    var IMS_Request_Type v_Request := p_IMS_DATA_REQ.Request;

    var template (value) ASP_DataRequest v_ASP_DataRequest;

    var template (omit) SigCompInfo v_SigCompInfo := omit;

    var MessageHeader v_MessageHeader;

    var boolean v_SecurityIsEstablished := vc_Global.SecurityStarted; //@sic R5s130195 - Change 2
    if (vc_Global.DLSigCompActive) {

      v_MessageHeader := f_IMS_Request_GetMessageHeader(v_Request);

      v_SigCompInfo := cs_SigCompDL(v_MessageHeader.callId.callid);

    }

    v_ASP_DataRequest := cas_ASP_DataRequest(v_SigCompInfo, fl_IP_ADAPTOR_GetPortInfo(v_RoutingInfo.Protocol), v_Request);

    if ((v_RoutingInfo.Security == protected) and (not v_SecurityIsEstablished)) {

        FatalError(__FILE__, __LINE__, "IMS security not established");

    }
    // Requests are sent on Client ports for UDP and TCP

    if (not v_SecurityIsEstablished) {  //@sic R5s130195 - Change 2
      UnsecClient.send(v_ASP_DataRequest);

    } else {

      SecClient.send(v_ASP_DataRequest);

    }

  }
Change 2.3

	Function name
	fl_SendResposetDL

	Reason for change
	Sending of responses needs to consider whether security is initated or not.

	Summary of change
	See below

	TTCN module
	IMS/34_229_Model/IP_ADAPTOR_Component.ttcn

	MCC160 Comment
	MCC160 implementation


Before change

  function fl_SendResponseDL(integer p_PdnIndex,

                             IMS_DATA_RSP p_IMS_DATA_RSP) runs on IP_ADAPTOR_PTC

  { /* !!!! NOTE: p_PdnIndex is currently not used but may be needed to distinguish SA for different PDNs !!!! */

    var IMS_RoutingInfo_Type v_RoutingInfo := p_IMS_DATA_RSP.RoutingInfo;

    var template (value) ASP_DataResponse v_ASP_DataResponse;

    var template (omit) SigCompInfo v_SigCompInfo := omit;

    if (vc_Global.DLSigCompActive) {

      v_SigCompInfo := cs_SigCompDL(p_IMS_DATA_RSP.Response.msgHeader.callId.callid);

    }

    v_ASP_DataResponse := cas_ASP_DataResponse(v_SigCompInfo, fl_IP_ADAPTOR_GetPortInfo(v_RoutingInfo.Protocol), p_IMS_DATA_RSP.Response);

    // Responses are sent on Server port for TCP and on Client port for UDP

    if (v_RoutingInfo.Protocol == tcp) {

      if (v_RoutingInfo.Security == unprotected) {
        UnsecServer.send(v_ASP_DataResponse);

      } else {

        SecServer.send(v_ASP_DataResponse);

      }

    } else {

      if (v_RoutingInfo.Security == unprotected) {

        UnsecClient.send(v_ASP_DataResponse);

      } else {

        SecClient.send(v_ASP_DataResponse);

      }

    }

  }

After change

  function fl_SendResponseDL(integer p_PdnIndex,

                             IMS_DATA_RSP p_IMS_DATA_RSP) runs on IP_ADAPTOR_PTC

  { /* !!!! NOTE: p_PdnIndex is currently not used but may be needed to distinguish SA for different PDNs !!!! */

    var IMS_RoutingInfo_Type v_RoutingInfo := p_IMS_DATA_RSP.RoutingInfo;

    var template (value) ASP_DataResponse v_ASP_DataResponse;

    var template (omit) SigCompInfo v_SigCompInfo := omit;

    var boolean v_SecurityIsEstablished := vc_Global.SecurityStarted; //@sic R5s130195 - Change 2
    if (vc_Global.DLSigCompActive) {

      v_SigCompInfo := cs_SigCompDL(p_IMS_DATA_RSP.Response.msgHeader.callId.callid);

    }

    v_ASP_DataResponse := cas_ASP_DataResponse(v_SigCompInfo, fl_IP_ADAPTOR_GetPortInfo(v_RoutingInfo.Protocol), p_IMS_DATA_RSP.Response);

    if ((v_RoutingInfo.Security == protected) and (not v_SecurityIsEstablished)) {

        FatalError(__FILE__, __LINE__, "IMS security not established");

    }

    // Responses are sent on Server port for TCP and on Client port for UDP

    if (v_RoutingInfo.Protocol == tcp) {

      if (not v_SecurityIsEstablished) {  //@sic R5s130195 - Change 2
        UnsecServer.send(v_ASP_DataResponse);

      } else {

        SecServer.send(v_ASP_DataResponse);

      }

    } else {

      if (not v_SecurityIsEstablished) {  //@sic R5s130195 - Change 2
        UnsecClient.send(v_ASP_DataResponse);

      } else {

        SecClient.send(v_ASP_DataResponse);

      }

    }

  }

Change 2.4

	Altstep  name
	a_IMS_PTC_Config

	Reason for change
	Security flag needs to be set when the IP sec configuration is activated

	Summary of change
	See below

	TTCN module
	IMS/34_229_Model/IP_ADAPTOR_Component.ttcn

	MCC160 Comment
	MCC160 implementation


Before change

  altstep a_IMS_PTC_Config(IP_IMS_CTRL_PORT p_Port,

                           integer p_PdnIndex) runs on IP_ADAPTOR_PTC

  { /* @sic R5s130266 change 1.3, 4, 5 - MCC160 Implementation: SPIs and protected ports are fully controlled by the IMS PTC sic@ */

    var IMS_CONFIG_REQ v_IMS_CONFIG_REQ;

    var IPsec_SecurityKeys_Type v_SecurityKeys;

    var IMS_PortsAndSecurityConfigReq_Type v_PortsAndSecurityConfig;

    var IMS_SecurityInfo_Type v_SecurityInfo;

    var IPsec_IntegrityAlgorithm_Type v_IntegrityAlgorithm;

    var IPsec_CipheringAlgorithm_Type v_CipheringAlgorithm;

    var IMS_RegistrationInfo_Type v_RegistrationInfo;

    var charstring v_UE_Address;

    var template (value) IPsec_SecurityAssociationList_Type v_SA_List;

    var integer v_Port_us;

    [] p_Port.receive(car_IMS_InstallKey_REQ) -> value v_IMS_CONFIG_REQ

      {

        v_SecurityKeys := v_IMS_CONFIG_REQ.InstallKey;

        fl_IP_ADAPTOR_SecurityKeyConfig(v_SecurityKeys.MD5_96Key,

                                        v_SecurityKeys.SHA_1_96Key,

                                        v_SecurityKeys.DES_EDE3_CBCKey,

                                        v_SecurityKeys.AES_CBCKey,

                                        p_PdnIndex);

        p_Port.send(cas_IMS_InstallKey_CNF);

      }

    [] p_Port.receive(car_IMS_PortsAndSecurityConfig_REQ) -> value v_IMS_CONFIG_REQ

      {

        v_PortsAndSecurityConfig := v_IMS_CONFIG_REQ.PortsAndSecurityConfig;

        v_RegistrationInfo := v_PortsAndSecurityConfig.RegistrationInfo;

        if ( ischosen(v_RegistrationInfo.UE_Address.V4)) {

          v_UE_Address := v_RegistrationInfo.UE_Address.V4.Addr;

        } else {

          v_UE_Address := v_RegistrationInfo.UE_Address.V6.Addr;

        }

        // Need to update the UE IP Address

        fl_IP_ADAPTOR_ConfigureIPAddr(px_P_CSCF_IPAddr,

                                      px_DNSServer_IPAddr,

                                      px_DHCPServer_IPAddr,

                                      v_UE_Address,

                                      px_PeerUE_IPAddr,

                                      p_PdnIndex);

        // Update the UE's port number

        // set port number:

        if (isvalue(v_PortsAndSecurityConfig.UnprotectedPort_us)) {

          v_Port_us := v_PortsAndSecurityConfig.UnprotectedPort_us;   // set UE's port number as requested by the UE

        } else {

          v_Port_us := tsc_IMS_PortNumber_5060;                       // restore port number e.g. if it has been set to other value in previous session

        }

        fl_IP_ADAPTOR_UpdatePCSCFRemotePort(v_Port_us, p_PdnIndex);

        if (isvalue(v_RegistrationInfo.SecurityInfo)) {                // non-GIBA

          v_SecurityInfo := v_RegistrationInfo.SecurityInfo;

          v_IntegrityAlgorithm := v_SecurityInfo.IntegrityAlgorithm;

          v_CipheringAlgorithm := v_SecurityInfo.CipheringAlgorithm;

          // configure SAs

          v_SA_List := f_IPsec_SA_List(v_RegistrationInfo.NW_Address, v_RegistrationInfo.UE_Address, v_SecurityInfo, v_IntegrityAlgorithm, v_CipheringAlgorithm);

          fl_IP_ADAPTOR_SAConfig(valueof(v_SA_List), p_PdnIndex);

        }

        p_Port.send(cas_IMS_PortsAndSecurityConfig_CNF);

2      

      }

  }
After change

    altstep a_IMS_PTC_Config(IP_IMS_CTRL_PORT p_Port,

                           integer p_PdnIndex) runs on IP_ADAPTOR_PTC

  { /* @sic R5s130266 change 1.3, 4, 5 - MCC160 Implementation: SPIs and protected ports are fully controlled by the IMS PTC sic@ */

    var IMS_CONFIG_REQ v_IMS_CONFIG_REQ;

    var IPsec_SecurityKeys_Type v_SecurityKeys;

    var IMS_PortsAndSecurityConfigReq_Type v_PortsAndSecurityConfig;

    var IMS_SecurityInfo_Type v_SecurityInfo;

    var IPsec_IntegrityAlgorithm_Type v_IntegrityAlgorithm;

    var IPsec_CipheringAlgorithm_Type v_CipheringAlgorithm;

    var IMS_RegistrationInfo_Type v_RegistrationInfo;

    var charstring v_UE_Address;

    var template (value) IPsec_SecurityAssociationList_Type v_SA_List;

    var integer v_Port_us;

    [] p_Port.receive(car_IMS_InstallKey_REQ) -> value v_IMS_CONFIG_REQ

      {

        v_SecurityKeys := v_IMS_CONFIG_REQ.InstallKey;

        fl_IP_ADAPTOR_SecurityKeyConfig(v_SecurityKeys.MD5_96Key,

                                        v_SecurityKeys.SHA_1_96Key,

                                        v_SecurityKeys.DES_EDE3_CBCKey,

                                        v_SecurityKeys.AES_CBCKey,

                                        p_PdnIndex);

        p_Port.send(cas_IMS_InstallKey_CNF);

      }

    [] p_Port.receive(car_IMS_PortsAndSecurityConfig_REQ) -> value v_IMS_CONFIG_REQ

      {

        v_PortsAndSecurityConfig := v_IMS_CONFIG_REQ.PortsAndSecurityConfig;

        v_RegistrationInfo := v_PortsAndSecurityConfig.RegistrationInfo;

        if ( ischosen(v_RegistrationInfo.UE_Address.V4)) {

          v_UE_Address := v_RegistrationInfo.UE_Address.V4.Addr;

        } else {

          v_UE_Address := v_RegistrationInfo.UE_Address.V6.Addr;

        }

        // Need to update the UE IP Address

        fl_IP_ADAPTOR_ConfigureIPAddr(px_P_CSCF_IPAddr,

                                      px_DNSServer_IPAddr,

                                      px_DHCPServer_IPAddr,

                                      v_UE_Address,

                                      px_PeerUE_IPAddr,

                                      p_PdnIndex);

        // Update the UE's port number

        // set port number:

        if (isvalue(v_PortsAndSecurityConfig.UnprotectedPort_us)) {

          v_Port_us := v_PortsAndSecurityConfig.UnprotectedPort_us;   // set UE's port number as requested by the UE

        } else {

          v_Port_us := tsc_IMS_PortNumber_5060;                       // restore port number e.g. if it has been set to other value in previous session

        }

        fl_IP_ADAPTOR_UpdatePCSCFRemotePort(v_Port_us, p_PdnIndex);

        if (isvalue(v_RegistrationInfo.SecurityInfo)) {                // non-GIBA

          v_SecurityInfo := v_RegistrationInfo.SecurityInfo;

          v_IntegrityAlgorithm := v_SecurityInfo.IntegrityAlgorithm;

          v_CipheringAlgorithm := v_SecurityInfo.CipheringAlgorithm;

          // configure SAs

          v_SA_List := f_IPsec_SA_List(v_RegistrationInfo.NW_Address, v_RegistrationInfo.UE_Address, v_SecurityInfo, v_IntegrityAlgorithm, v_CipheringAlgorithm);

          fl_IP_ADAPTOR_SAConfig(valueof(v_SA_List), p_PdnIndex);

        }

        p_Port.send(cas_IMS_PortsAndSecurityConfig_CNF);

        // Finally set the flag indicating start of security

        vc_Global.SecurityStarted := true;  //@sic R5s130195 - Change 2      

      }

  }
Change 3 – a_SystemInterfaceHandler
	Altstep name
	a_SystemInterfaceHandler

	Reason for change
	Allow to receive TCP requests on unsecure port

	Summary of change
	Deleted clause that disallows TCP on unsecure port for responses

	TTCN module
	IMS/34229_Model/IP_ADAPTOR_Component

	MCC160 Comment
	Accepted


Before change

  altstep a_SystemInterfaceHandler() runs on IP_ADAPTOR_PTC

  { /* NOTE 1: we don't need a default handler for this PTC since this altsteps shall handle all ports and messages

     * NOTE 2: any messages at the configurations ports (IpConf, SigComp, mmiPort) shall be responses to a request i.e. we await this explicitly after the request has been sent */

    var ASP_DataResponse v_ASP_DataResponse;

    var ASP_DataRequest v_ASP_DataRequest;

    var SSPortInfo v_SSPortInfo;

    var integer v_PdnIndex;

    // Client ports

    [] UnsecClient.receive(car_ASP_DataResponse_Any) -> value v_ASP_DataResponse {

      v_SSPortInfo := v_ASP_DataResponse.portInfo;

      if (v_SSPortInfo.ipAddr == px_UE_IPAddr2) {

        v_PdnIndex := tsc_Index_PDN2;

      } else {

        v_PdnIndex := tsc_Index_PDN1;

      }

      fl_SendResponseUL(v_PdnIndex, unprotected, v_ASP_DataResponse);

    }

    [] SecClient.receive(car_ASP_DataResponse_Any) -> value v_ASP_DataResponse {

      v_SSPortInfo := v_ASP_DataResponse.portInfo;

      if (v_SSPortInfo.ipAddr == px_UE_IPAddr2) {

        v_PdnIndex := tsc_Index_PDN2;

      } else {

        v_PdnIndex := tsc_Index_PDN1;

      }

      fl_SendResponseUL(v_PdnIndex, protected, v_ASP_DataResponse);

    }

    [] UnsecClient.receive(car_ASP_DataRequest_Any) {

      fl_IP_ADAPTOR_Error(__FILE__, __LINE__, "unexpected data request at port 'Client'");

    }

    [] SecClient.receive(car_ASP_DataRequest_Any) {

      fl_IP_ADAPTOR_Error(__FILE__, __LINE__, "unexpected data request at port 'SecClient'");

    }

    // Server ports

    [] UnsecServer.receive(car_ASP_DataRequest_Any) -> value v_ASP_DataRequest {

      v_SSPortInfo := v_ASP_DataRequest.portInfo;

      if (v_SSPortInfo.ipAddr == px_UE_IPAddr2) {

        v_PdnIndex := tsc_Index_PDN2;

      } else {

        v_PdnIndex := tsc_Index_PDN1;

      }

      fl_SendRequestUL(v_PdnIndex, unprotected, v_ASP_DataRequest);

    }

    [] SecServer.receive(car_ASP_DataRequest_Any) -> value v_ASP_DataRequest {

      v_SSPortInfo := v_ASP_DataRequest.portInfo;

      if (v_SSPortInfo.ipAddr == px_UE_IPAddr2) {

        v_PdnIndex := tsc_Index_PDN2;

      } else {

        v_PdnIndex := tsc_Index_PDN1;

      }

      fl_SendRequestUL(v_PdnIndex, protected, v_ASP_DataRequest);

    }

    [] UnsecServer.receive(car_ASP_DataResponse_Any) -> value v_ASP_DataResponse {   // in case of UDP any response is sent to Server port
      if (v_ASP_DataResponse.portInfo.transportProtocol != udp) {

        fl_IP_ADAPTOR_Error(__FILE__, __LINE__, "unexpected data response at port 'Server'");

      } else {
      v_SSPortInfo := v_ASP_DataResponse.portInfo;

      if (v_SSPortInfo.ipAddr == px_UE_IPAddr2) {

        v_PdnIndex := tsc_Index_PDN2;

      } else {

        v_PdnIndex := tsc_Index_PDN1;

      }

        fl_SendResponseUL(v_PdnIndex, unprotected, v_ASP_DataResponse);

      }
    }

    [] SecServer.receive(car_ASP_DataResponse_Any) -> value v_ASP_DataResponse {

      if (v_ASP_DataResponse.portInfo.transportProtocol != udp) {

        fl_IP_ADAPTOR_Error(__FILE__, __LINE__, "unexpected data response at port 'SecServer'");

      } else {

          v_SSPortInfo := v_ASP_DataResponse.portInfo;

          if (v_SSPortInfo.ipAddr == px_UE_IPAddr2) {

            v_PdnIndex := tsc_Index_PDN2;

          } else {

            v_PdnIndex := tsc_Index_PDN1;

          }

        fl_SendResponseUL(v_PdnIndex, protected, v_ASP_DataResponse);

      }

    }

    // config ports

    [] IpConf.receive {

      fl_IP_ADAPTOR_Error(__FILE__, __LINE__, "unexpected messsage at port 'IpConf'");   // see NOTE 2

    }

    [] SigComp.receive {

      fl_IP_ADAPTOR_Error(__FILE__, __LINE__, "unexpected messsage at port 'SigComp'");  // see NOTE 2

    }

  }
After change

  altstep a_SystemInterfaceHandler() runs on IP_ADAPTOR_PTC

  { /* NOTE 1: we don't need a default handler for this PTC since this altsteps shall handle all ports and messages

     * NOTE 2: any messages at the configurations ports (IpConf, SigComp, mmiPort) shall be responses to a request i.e. we await this explicitly after the request has been sent */

    var ASP_DataResponse v_ASP_DataResponse;

    var ASP_DataRequest v_ASP_DataRequest;

    var SSPortInfo v_SSPortInfo;

    var integer v_PdnIndex;

    // Client ports

    [] UnsecClient.receive(car_ASP_DataResponse_Any) -> value v_ASP_DataResponse {

      v_SSPortInfo := v_ASP_DataResponse.portInfo;

      if (v_SSPortInfo.ipAddr == px_UE_IPAddr2) {

        v_PdnIndex := tsc_Index_PDN2;

      } else {

        v_PdnIndex := tsc_Index_PDN1;

      }

      fl_SendResponseUL(v_PdnIndex, unprotected, v_ASP_DataResponse);

    }

    [] SecClient.receive(car_ASP_DataResponse_Any) -> value v_ASP_DataResponse {

      v_SSPortInfo := v_ASP_DataResponse.portInfo;

      if (v_SSPortInfo.ipAddr == px_UE_IPAddr2) {

        v_PdnIndex := tsc_Index_PDN2;

      } else {

        v_PdnIndex := tsc_Index_PDN1;

      }

      fl_SendResponseUL(v_PdnIndex, protected, v_ASP_DataResponse);

    }

    [] UnsecClient.receive(car_ASP_DataRequest_Any) {

      fl_IP_ADAPTOR_Error(__FILE__, __LINE__, "unexpected data request at port 'Client'");

    }

    [] SecClient.receive(car_ASP_DataRequest_Any) {

      fl_IP_ADAPTOR_Error(__FILE__, __LINE__, "unexpected data request at port 'SecClient'");

    }

    // Server ports

    [] UnsecServer.receive(car_ASP_DataRequest_Any) -> value v_ASP_DataRequest {

      v_SSPortInfo := v_ASP_DataRequest.portInfo;

      if (v_SSPortInfo.ipAddr == px_UE_IPAddr2) {

        v_PdnIndex := tsc_Index_PDN2;

      } else {

        v_PdnIndex := tsc_Index_PDN1;

      }

      fl_SendRequestUL(v_PdnIndex, unprotected, v_ASP_DataRequest);

    }

    [] SecServer.receive(car_ASP_DataRequest_Any) -> value v_ASP_DataRequest {

      v_SSPortInfo := v_ASP_DataRequest.portInfo;

      if (v_SSPortInfo.ipAddr == px_UE_IPAddr2) {

        v_PdnIndex := tsc_Index_PDN2;

      } else {

        v_PdnIndex := tsc_Index_PDN1;

      }

      fl_SendRequestUL(v_PdnIndex, protected, v_ASP_DataRequest);

    }

    [] UnsecServer.receive(car_ASP_DataResponse_Any) -> value v_ASP_DataResponse {   

      v_SSPortInfo := v_ASP_DataResponse.portInfo;

      if (v_SSPortInfo.ipAddr == px_UE_IPAddr2) {

        v_PdnIndex := tsc_Index_PDN2;

      } else {

        v_PdnIndex := tsc_Index_PDN1;

      }

        fl_SendResponseUL(v_PdnIndex, unprotected, v_ASP_DataResponse);

    }

    [] SecServer.receive(car_ASP_DataResponse_Any) -> value v_ASP_DataResponse {

      if (v_ASP_DataResponse.portInfo.transportProtocol != udp) {

        fl_IP_ADAPTOR_Error(__FILE__, __LINE__, "unexpected data response at port 'SecServer'");

      } else {

          v_SSPortInfo := v_ASP_DataResponse.portInfo;

          if (v_SSPortInfo.ipAddr == px_UE_IPAddr2) {

            v_PdnIndex := tsc_Index_PDN2;

          } else {

            v_PdnIndex := tsc_Index_PDN1;

          }

        fl_SendResponseUL(v_PdnIndex, protected, v_ASP_DataResponse);

      }

    }

    // config ports

    [] IpConf.receive {

      fl_IP_ADAPTOR_Error(__FILE__, __LINE__, "unexpected messsage at port 'IpConf'");   // see NOTE 2

    }

    [] SigComp.receive {

      fl_IP_ADAPTOR_Error(__FILE__, __LINE__, "unexpected messsage at port 'SigComp'");  // see NOTE 2

    }

  } 
